NASA Technical Memorandum 80179

Computational Methods for Analyzing the Transmission Characteristics of a Beta Particle Magnetic Analysis System

Jag J. Singh

DECEMBER 1979

NASA

NASA Technical Memorandum 80179

Computational Methods for Analyzing the Transmission Characteristics of a Beta Particle Magnetic Analysis System

Jag J. Singh
Langley Research Center
Hampton, Virginia

NASA
National Aeronautics
and Space Administration

Scientific and Technical
Information Branch

1979

SUMMARY

Computational methods have been developed to study the trajectories of beta particles (positrons) through a magnetic analysis system as a function of the spatial distribution of the radionuclides in the beta source, size, and shape of the source collimator, and the strength of the analyzer magnetic field. On the basis of these methods, the particle flux, their energy spectrum, and source-to-target transit times have been calculated for Na^{22} positrons as a function of the analyzer magnetic field and the size and location of the target. These data will be useful in studies requiring parallel beams of positrons of uniform energy -- such as measurement of the moisture distribution in composite materials.

INTRODUCTION

Recent studies (refs. 1 to 4) have shown that positron lifetime in composite materials is a linear function of the moisture content of the material up to the saturation level. This linear dependence of positron lifetime on moisture content is now being utilized for determining the moisture distribution profile in the polymeric targets. Moisture distribution profile is an important factor in determining the environmental degradation of composites. It is believed (ref. 5) that nonuniform moisture distribution leads to higher internal fatigue of these materials. Ability to measure moisture profile nondestructively is of interest since continual absorption and desorption of moisture from composites in service usually leaves the moisture profile highly nonuniform inside the test specimen.

In the previously completed studies (ref. 4), moisture profiling has been accomplished by using a Na^{22} positron source in conjunction with thin steel beam energy degraders. This process is necessarily gross since the positron spectrum is continuous in energy. It would be preferable to use positron beams of well-defined energy to determine moisture content at the end of their range in the test specimen. One practical technique of doing so would require the use of a magnetic analyzer for selecting positrons of predetermined energy to probe the moisture content at different depths in the test specimen. This paper describes analytical methods for maximizing the number of energy-analyzed positrons arriving at the target, while minimizing the spread in their energy and arrival time. Use of magnetically analyzed positrons would be of practical interest only if the spread in their arrival time at the target is less than the time resolution of the lifetime measurement system (refs. 1 and 2) and their energy resolution is better than their range definition in the target. The computer programs needed to calculate various experimental parameters for optimizing the experimental methods have been developed by Gerald H. Mall and are included as an appendix to this paper.

SYMBOLS

c	velocity of light
D	distance of center of the magnetic field circle from center of source
D_c	diameter of collimator
$E_{\beta+}$	kinetic energy of positron
$F(Z,W)$	Fermi function for positrons of energy W
L_c	length of collimator
ℓ	particle's path length to target
m	slope of the tangent to particle's path at point of exit from magnetic field
m_o	particle's rest mass
m^*	particle's relativistic mass
$N(W)\ dW$	number of positrons in the energy range from W to $W + dW$
P	transition matrix element for positron decay
R	radius of target
R_c	radius of collimator
R_F	radius of magnetic field region
R_p	radius of particle's circular path in the magnetic field
v	velocity of particle
W	total positron energy in units of m_oc^2
W_o	maximum value of W, corresponding to end-point beta-spectrum energy
x,y	coordinates of particle; the origin of the x-y coordinate system lies at the center of the source spot, with the x-axis passing through the center of the magnetic field region
Y_c	distance of center of magnetic field from target
θ	angle of particle emission from source, relative to x-axis

τ transit time of particle

τ_o universal time constant

ϕ angle subtended by particle's circular path in magnetic field at circular path center

PROBLEM STATEMENT

A well-collimated beam of Na^{22} positrons is allowed to enter a variable magnetic field. The purpose of this analysis is to follow the trajectories of positrons that escape the collimator and enter the magnetic field. The trajectory of each positron is followed until it arrives at the target located at different distances from the center of the magnetic field set to transmit preselected positron energies. Targets of sizes ranging in diameter from 1 cm to 2.54 cm, located at a number of distances from the center of the magnetic field, are selected for measuring the positron beam. Since the positron emission from the Na^{22} source nuclei is spherically symmetrical, the calculations are considered in a two-dimensional geometry, with the results generalized for a three-dimensional geometry. From the known energy spectrum of Na^{22} positrons (ref. 6), total transit times from the source to the target are calculated for each positron arriving at the target. Also, the energy spectrum of the positrons received at the target location for various settings of the magnetic field is determined to assess the definition of the range of the arriving positrons. Finally, the trajectories of all the positrons that are transmitted through preselected magnetic fields are followed for a distance of 20.0 cm from the center of the magnetic field. From these data, a distance at which the positron energy dispersion and the transit time spread are the smallest and the positron number density is the highest is determined for optimal location of the target.

THEORETICAL CALCULATIONS

For the purpose of developing an analytical basis for the computational methods, it is assumed that the positron source is deposited at the bottom of a long narrow cylinder. The open end of the cylinder serves to collimate the positron beam entering the magnetic field. The source profile is assumed to be one of the three practically attainable configurations illustrated in figure 1. The geometrical dimensions of the source collimator considered are shown in figure 2. Figure 3 shows a schematic diagram of the experimental system.

Particle's Trajectory Through Source Collimator

As depicted in the following sketch, the particle is emitted at point $(0,y_i)$ at an angle θ relative to the x-axis with a velocity v:



The path of the particle is described by

$$x = vt \cos \theta \tag{1a}$$

$$y = vt \sin \theta + y_i \tag{1b}$$

Solving equation (1a) for the product vt corresponding to $x = L_c$ yields

$$vt|_{x=L_c} = \frac{L_c}{\cos \theta} \tag{2}$$

Substituting equation (2) into equation (1b) leads to

$$y_E = y|_{x=L_c} = L_c \tan \theta + y_i \tag{3}$$

If $y_E \geqq R_c$ or if $y_E \leqq -R_c$, the particle strikes the wall of the collimator and is not considered further. If $-R_c < y_E < R_c$, the intersection of the trajectory with the circular magnetic field is calculated.

First Intersection of Particle's Trajectory With Magnetic Field

In the following sketch, the particle's trajectory is represented by the line from $(0,y_i)$ to (x_1,y_1), and the magnetic field by the circle:



The equations of the line and the circle are given by

$$y = x \tan \theta + y_i \tag{4a}$$

$$\left[x - (R_F + L_c)\right]^2 + y^2 = R_F^2 \tag{4b}$$

Substituting y from equation (4a) into equation (4b) yields

$$\left[x_1 - (R_F + L_c)\right]^2 + (x_1 \tan \theta + y_i)^2 = R_F^2 \tag{5}$$

Expanding equation (5) and making the substitution $R_F + L_c = D$ leads to

$$x_1^2(1 + \tan^2 \theta) + x_1(2y_i \tan \theta - 2D) + (D^2 + y_i^2 - R_F^2) = 0 \tag{6}$$

Equation (6) may be solved for x_1:

$$x_1 = \frac{(2D - 2y_i \tan \theta) - \sqrt{(2y_i \tan \theta - 2D)^2 - 4(1 + \tan^2 \theta)(D^2 + y_i^2 - R_F^2)}}{2(1 + \tan^2 \theta)} \tag{7}$$

where the smaller root was selected by subtracting the square root. Expanding equation (7), collecting terms, and reducing the results leads to

$$x_1 = \frac{(D - y_i \tan\theta) - \sqrt{R_F^2 \sec^2\theta - (y_i + D\tan\theta)^2}}{\sec^2\theta} \tag{8}$$

Equation (8) gives the x-coordinate of the desired intersection point and the y-coordinate may be obtained by substituting equation (8) into equation (4a).

Second Intersection of Particle's Trajectory With Magnetic Field

Upon entering the magnetic field (assumed to be perpendicular to the plane of the paper), the particle experiences forces causing it to move in a circular path.



The forces experienced by the particle satisfy

$$qvH = m^*v^2/R_p \tag{9}$$

where q is the charge of the particle, m^* is the particle's mass at v, and H is the magnitude of the magnetic field. The radius of the particle's circular path is then given by

$$R_p = m^*v/qH \tag{10}$$

and the tangent to the path at the point (x_1,y_1) is the slope of the initial trajectory, $\tan\theta$. The equations of the two circles are given by

$$(x - D)^2 + y^2 = R_F^2 \tag{11a}$$

$$(x - x_p)^2 + (y - y_p)^2 = R_p^2 \tag{11b}$$

Solving equation (11a) for y and substituting into equation (11b) produces the following equation for x_2:

$$(x_2 - x_p)^2 + \left[\pm\sqrt{R_F^2 - (x_2 - D)^2} - y_p\right]^2 = R_p^2 \tag{12}$$

Expanding equation (12) leads to

$$x_2^2 - 2x_2x_p + x_p^2 + R_F^2 - (x_2 - D)^2 \mp 2y_p\sqrt{R_F^2 - (x_2 - D)^2} + y_p^2 = R_p^2 \tag{13}$$

By making the substitutions $K = 2D - 2x_p$, $L = x_p^2 + y_p^2 + R_F^2 - R_p^2 - D^2$, equation (13) can be expressed as

$$Kx_2 + L = \pm\, 2y_p\sqrt{R_F^2 - (x_2 - D)^2} \tag{14}$$

Squaring both sides of equation (14) and collecting terms leads to the following expression for x_2:

$$Sx_2^2 + Tx_2 + U = 0 \tag{15}$$

where $S = K^2 + 4y_p^2$, $T = 2KL - 8y_p^2D$, and $U = L^2 + 4y_p^2(D^2 - R_F^2)$. Then, x_2 is given the following equation where the positive sign of the square root has been selected:

$$x_2 = \frac{-T + \sqrt{T^2 - 4SU}}{2S} \tag{16}$$

The corresponding value for y_2 may be determined by substituting equation (16) into equation (11a).

Intersection of Particle's Trajectory With Line $x = L_c + R_F$

The following sketch illustrates the particle's trajectory after leaving the magnetic field:



Equation (11b) may be solved for y and differentiated with respect to x to find the slope of the tangent to the particle's circular path at (x_2,y_2). The result is

$$m = \frac{x_2 - x_p}{\sqrt{R_p^2 - (x_2 - x_p)^2}} \tag{17}$$

The point of intersection of the particle's trajectory with the line $x = L_C + R_F$ is then given by

$$x_3 = L_C + R_F \tag{18a}$$

$$y_3 = y_2 - m(x_2 - L_C - R_F) \tag{18b}$$

Determination of Particle's Transit Time

The particle trajectory along the circular path traverses an angle ϕ, given by

$$\phi = \cos^{-1}\left[\frac{2R_p^2 - (x_2 - x_1)^2 - (y_2 - y_1)^2}{2R_p^2}\right] \tag{19}$$

The total path length is then given by the following expression:

$$\ell = R_p\phi + \sqrt{(x_1 - x_i)^2 + (y_1 - y_i)^2} + \sqrt{(x_3 - x_2)^2 + (y_3 - y_2)^2} \tag{20}$$

For a particle with kinetic energy T, the relativistic mass is given by

$$m^* = \frac{T + m_o c^2}{c^2} \tag{21}$$

where m_o is the rest mass and c is the velocity of light. The particle's velocity is given by the following equation:

$$v = \sqrt{1 - (m_o/m^*)^2}\; c \tag{22}$$

Equations (20) and (22) can be combined to yield the transit time:

$$\tau = \ell/v \tag{23}$$

Using equations developed in these sections the various parameters of interest have been calculated for a source, collimator, and magnetic field geometry of the type shown in figures 1 and 4, 2, and 3, respectively. The results are discussed in the following section.

DISCUSSION OF TYPICAL RESULTS

Trajectories of positrons for three different source configurations, transmitted through the magnetic analysis system set for 200 keV, are shown in figures 5 to 7. It is clear that none of the source configurations considered lead to actual positron focusing at distances up to 20 cm from the center of the magnetic field. The flux densities for the flat and convex sources were approximately equal and considerably higher than that for the concave source. Also the convex source provided slightly better energy and time resolution than the other configurations. It was therefore decided to use the convex source configuration (fig. 4) for detailed calculations of the transmitted positron beam parameters. Some typical results for the spectra of positrons from a convex source arriving at several target locations for two different size targets are shown in figures 8 to 10 and are summarized in tables I to III. These results include the effects of Na^{22} source deposit profile as well as the positron Coulomb correction factor (refs. 7 to 9). The number of positrons emitted at the source has been taken as 1000 in each case.

It is clear from the tables that for a 1-cm-diameter target located a distance from the magnetic field center $Y_c \geqq 12$ cm, the dispersion in the arriving positron energy is less than 10% of the energy for which the magnetic field is set for all cases; that is, $\Delta E/E < 10\%$. Furthermore, the spread in transit time $\Delta\tau$ for positrons arriving at the 1-cm-diameter target is less than 100 ps for $Y_c \geqq 12$ cm. This transit time spread is less than the time resolution of the lifetime measurement system. For a 2.54-cm-diameter target, $\Delta E/E \approx 10\%$ and $\Delta\tau < 100$ ps for $Y_c \geqq 20$ cm for all magnetic field settings. Thus a target to magnetic field center distance of approximately 20 cm appears to be acceptable for both the 1-cm and the 2.54-cm targets. This conclusion is also corroborated by the data shown in figure 11. This figure shows the trajectories of all the positrons transmitted to the target through the magnetic analyzer when the latter is set for $E_{\beta+} = 300$ keV for 1-cm and 2.54-cm targets. (Similar results are obtained when the magnetic field is set for other positron energies.)

From the foregoing discussion, it is seen that for the 1-cm target, target to magnetic field center distances greater than 12 cm would be acceptable, though 12 cm would yield the highest number density at the target. For the 2.54-cm target, the minimum acceptable distance is 20 cm.

CONCLUSIONS

A computer program for optimizing a positron magnetic analysis system for studying moisture profile in composites has been developed. Three types of source configurations were considered. It appears that a convex source deposit provides simultaneously the smallest energy and transit time spread and the largest positron flux density at the target locations up to 20 cm. For such a source, the energy dispersion in the positrons arriving at a 1-cm-diameter target located at or over 12 cm from the magnetic field center is less than 10% for all settings of the analyzing magnetic field. The transit time spread in the positrons striking the target under these conditions is less than 100 ps, which is less than the time resolution of the positron lifetime measurement system. An examination of the trajectories of the positrons from a convex source transmitted through the magnetic analyzer for several field settings confirms that a separation distance of approximately 12 cm also corresponds to the highest positron density in the transmitted beam. For a 2.54-cm-diameter target, the optimum distance would be 20 cm for the smallest energy and transit time spread and the largest positron number density at the target.

Langley Research Center
National Aeronautics and Space Administration
Hampton, VA 23665
November 20, 1979

APPENDIX

COMPUTER PROGRAMS FOR CALCULATING POSITRON TRANSMISSION THROUGH A MAGNETIC ANALYSIS SYSTEM

Gerald H. Mall
Computer Sciences Corporation
Hampton, Virginia

PROGRAM FOR CALCULATING ENERGY AND TRANSIT TIMES OF POSITRONS

Program Description

The computer program POSTRN is written in FORTRAN IV language for the Control Data CYBER 170 series digital computer system with network operating system (NOS) 1.2. The program including interactive graphics routines requires 60700 octal locations of core storage. A typical case in which 5000 positron trajectories are computed requires approximately 20 CPU seconds on the CYBER 173. Although the program has been designed to be used interactively, it can also be executed in a batch mode.

The program may be executed in three distinct ways. When the target location, target size, magnetic field setting, and desired number of trajectories are specified, the program uses a random number generator and two input probability distribution tables to determine the initial conditions of each positron. Programs have been developed to generate these tables giving probability versus energy and probability versus positron position within the collimator, and these programs are described subsequently in this appendix. When the program is executed in this manner, the output consists of several parameters describing those positrons which strike the target. Additionally, two disk files (TAPE1 and TAPE2) are generated which contain sufficient data to reconstruct the trajectory of each positron. TAPE1 contains data for positrons which miss the target and TAPE2 contains data for positrons which hit the target. The second mode of operation of the program requires as input one of the disk files TAPE1 or TAPE2. When this option is selected, program output consists of a plot of the trajectories of a specified number of positrons which have trajectory data on the disk file. The third mode of operation is similar to the second except that the user must provide trajectory data for each positron which is to be plotted. These data include the magnetic field setting and the energy, position within the collimator, and emission angle of each positron.

Description of FORTRAN Variables

The following list contains a description of the significant FORTRAN variables appearing in the program. The dimension for each array is beside the variable in parentheses. Each variable is also identified as I, input variable; P, program variable; or O, output variable. Variables which may be input or program variables, depending on the mode of operation, are identified by I/P.

FORTRAN variable	Type	Description
COLL	P	Collimator length
COLR	P	Collimator radius
EFLD	I	Energy for which the magnetic field is set
EMAX	O	Maximum energy of those positrons striking the target
EMIN	O	Minimum energy of those positrons striking the target
ENER	I/P	Positron energy
MODE	I	Mode of program operation
NHIT	O	Number of positrons which strike the target
NMISS	O	Number of positrons which miss the target
NTRY	I	Number of trajectories to be computed
RC	I	Target radius
RM	P	Magnetic field radius
THETA	I/P	Angle of emitted positron
TMAX	O	Maximum transit time of those positrons striking the target
TMIN	O	Minimum transit time of those positrons striking target
VEL	P	Positron velocity
X(1000)	P	x-coordinate of points to be plotted
Y(1000)	P	y-coordinate of points to be plotted
YC	I	Distance of target from center of magnetic field
YP	I/P	Position of emitted positron within the collimator

Flow Charts and Listings of Program

Program POSTRN.- POSTRN, the main program, performs all input and output operations, controls flow through the kinematics and geometry subroutines, and calls those subroutines which load the probability tables or generate graphic output depending on the specified mode of program operation. The flow chart and listing of POSTRN follow.



APPENDIX


1
Call PSEUDO
Call INIT
Read field setting
Call KINE
MODE = 1
T
F
Read trajectory data from TAPE7
Read trajectory data from INPUT
Call KINE
Call GEOM
Call PLT
Finished all trajectories
F
T
Call NFRAME
Call CALPLT
STOP

APPENDIX

```
      PROGRAM POSTRN(OUTPUT,INPUT,TAPE5=INPUT,TAPE6=OUTPUT,
     1 TAPE1,TAPE2,TAPE3,TAPE4,TAPE7)
      DIMENSION X(1000),Y(1000)
      COMMON X,Y,ENER,XMAS,VEL,H,RM,RP,COLL,COLR,YP,THETA,
     1 XP,X1,Y1,X2,Y2,XCM,YCM,XCP,YCP,X3,Y3,PL,TIME,IJK,
     2 EMIN,EMAX,NHIT,YC,RC,EFLD,TMIN,TMAX,NMISS,MODE
      DIMENSION EEE(100),YYY(100),TTT(100)
      COMMON/JTB/NFR,JREQ,IBAUD,JCCS,IJO,TFAC,IJTB(4)
      NFR = 0
      JREQ = 0
      IBAUD = 120
      JCCS = 0
      TFAC = 0.016
      READ(5,*) MODE
      IF(MODE.NE.0) GO TO 30
      CALL INIT
      READ(5,*) EFLD,YC,RC,NTRY
      IJK = 1
      ENER = EFLD
      CALL KINE
      DO 20 I=1,NTRY
      CALL DISTR
      IJK = I + 1
      CALL KINE
      CALL GEOM
      IF(X(1).LT.0.) GO TO 10
      CALL PRT
      GO TO 20
   10 NMISS = NMISS + 1
   20 CONTINUE
      WRITE(6,1) NHIT,NMISS,NTRY,EMIN,EMAX,TMIN,TMAX
    1 FORMAT(1H1,I10,* HITS*,I10,* MISSES*,I10,* TRYS*/
     1 1X,*MINIMUM ENERGY = *,E16.8,5X,*MAXIMUM ENERGY = *,E16.8/
     2 1X,*MINIMUM TIME = *,E16.8,5X,*MAXIMUM TIME = *,E16.8)
      STOP
   30 CONTINUE
      CALL PSEUDO
      CALL INIT
      READ(5,*) NTRY,EFLD
      ENER = EFLD
      IJK = 1
      CALL KINE
      DO 40 I=1,NTRY
      IF(MODE.EQ.1) READ(7) EEE(I),YYY(I),TTT(I)
      IF(MODE.EQ.2) READ(5,*) EEE(I),YYY(I),TTT(I)
   40 CONTINUE
      DO 50 I=1,NTRY
      IJK = I + 1
      ENER = EEE(I)
      YP = YYY(I)
      THETA = TTT(I)
      CALL KINE
      CALL GEOM
      IF(X(1).LT.0.) GO TO 50
      CALL PLT
   50 CONTINUE
      CALL NFRAME
      CALL CALPLT(0.,0.,999)
      STOP
      END
```

APPENDIX

Subroutine INIT.- Subroutine INIT initializes the dimensions of the collimator and the magnetic field. Several variables used in the output summary are also initialized.




```
      SUBROUTINE INIT
      DIMENSION X(1000),Y(1000)
      COMMON X,Y,ENER,XMAS,VEL,H,RM,RP,COLL,COLR,YP,THETA,
     1 XP,X1,Y1,X2,Y2,XCM,YCM,XCP,YCP,X3,Y3,PL,TIME,IJK,
     2 EMIN,EMAX,NHIT,YC,RC,EFLD,TMIN,TMAX,NMISS,MODE
      EMIN = 600.
      EMAX = 0.
      TMIN = 100.
      TMAX = 0.
      NMISS = 0
      NHIT = 0
      RM = 3.48*2.54
      RP = RM
      COLL = 3.*2.54
      COLR = 0.152*2.54/2.
      RETURN
      END
```

APPENDIX

<u>Subroutine KINE</u>.- Subroutine KINE computes the velocity of a relativistic positron of known energy. On the first call, the magnetic field strength required to focus positrons of the specified energy is determined. On subsequent calls, the radius of each positron's trajectory through the computed magnetic field is calculated.




```
      SUBROUTINE KINE
      DIMENSION X(1000),Y(1000)
      COMMON X,Y,ENER,XMAS,VEL,H,RM,RP,COLL,COLR,YP,THETA,
     1 XP,X1,Y1,X2,Y2,XCM,YCM,XCP,YCP,X3,Y3,PL,TIME,IJK,
     2 EMIN,EMAX,NHIT,YC,RC,EFLD,TMIN,TMAX,NMISS,MODE
      T = 1.6E-16*ENER
      XMAS = (T + 9.E+16*9.11E-31)/9.E+16
      TEMP = (9.11E-31/XMAS)**2
      TEMP = SQRT(1. - TEMP)
      VEL = TEMP*3.E+08
      HR = XMAS*VEL/1.6E-19
      HR = 1.E+06*HR
      XMAS = 1000.*XMAS
      VEL = 100.*VEL
      IF(IJK.EQ.1) H = HR/RP
      IF(IJK.NE.1) RP = HR/H
      RETURN
      END
```

APPENDIX

Subroutine DISTR.- Subroutine DISTR is the executive routine for assigning an energy and position within the collimator to each positron subject to user-defined probability distributions. An emission angle is also assigned subject to a uniform distribution $-\tan^{-1}(2 \cdot COLR/COLL) \leq \theta \leq \tan^{-1}(2 \cdot COLR/COLL)$.




```
      SUBROUTINE DISTR
      DIMENSION X(1000),Y(1000)
      COMMON X,Y,ENER,XMAS,VEL,H,RM,RP,COLL,COLR,YP,THETA,
     1 XP,X1,Y1,X2,Y2,XCM,YCM,XCP,YCP,X3,Y3,PL,TIME,IJK,
     2 EMIN,EMAX,NHIT,YC,RC,EFLD,TMIN,TMAX,NMISS,MODE
      CALL EDST
      CALL POSDST
      PI = 0.0506145
      TMX = -PI
      THETA = TMX + 2.*PI*RANF(DUM)
      RETURN
      END
```

APPENDIX

Subroutine EDST.- Subroutine EDST assigns a random energy to each positron subject to a user-specified probability distribution. On the first entry the subroutine reads from TAPE3 an array of energies and a corresponding array of normalized cumulative probabilities. A program which generates these arrays is described subsequently in this appendix. A uniform random number between 0 and 1 is used to interpolate linearly and select an energy. On subsequent entries, the read operation is bypassed.



APPENDIX

```
      SUBROUTINE EDST
      DIMENSION X(1000),Y(1000)
      COMMON X,Y,ENER,XMAS,VEL,H,RM,RP,COLL,COLR,YP,THETA,
     1 XP,X1,Y1,X2,Y2,XCM,YCM,XCP,YCP,X3,Y3,PL,TIME,IJK,
     2 EMIN,EMAX,NHIT,YC,RC,EFLD,TMIN,TMAX,NMISS,MODE
      DIMENSION XX(300),YY(300)
      DATA NIN/0/
      IF(NIN.NE.0) GO TO 30
      K = 1
      NIN = NIN + 1
      REWIND 3
   10 CONTINUE
      READ(3,1) XX(K),YY(K)
    1 FORMAT(2E16.8)
      IF(EOF(3).NE.0) GO TO 20
      K = K + 1
      GO TO 10
   20 K = K - 1
   30 KM1 = K - 1
   40 T = RANF(DUM)
      IF(T.LT.YY(1)) GO TO 40
      DO 50 I=1,KM1
      IF(T.GE.YY(I).AND.T.LE.YY(I+1)) GO TO 60
   50 CONTINUE
      I = KM1
   60 ENER = XX(I) + (XX(I+1)-XX(I))*(T - YY(I))/
     1 (YY(I+1) - YY(I))
      RETURN
      END
```

APPENDIX

Subroutine POSDST.- Subroutine POSDST assigns a random position within the collimator to each positron subject to a user-specified probability distribution. On the first entry the subroutine reads from TAPE4 an array of positions and a corresponding array of normalized cumulative probabilities. A program which generates these arrays is described subsequently in this appendix. A uniform random number between 0 and 1 is used to interpolate linearly and select a position. On subsequent entries, the read operation is bypassed.



APPENDIX

```
      SUBROUTINE POSDST
      DIMENSION X(1000),Y(1000)
      COMMON X,Y,ENER,XMAS,VEL,H,RM,RP,COLL,COLR,YP,THETA,
     1 XP,X1,Y1,X2,Y2,XCM,YCM,XCP,YCP,X3,Y3,PL,TIME,IJK,
     2 EMIN,EMAX,NHIT,YC,RC,EFLD,TMIN,TMAX,NMISS,MODE
      DIMENSION XX(300),YY(300)
      DATA NIN/0/
      IF(NIN.NE.0) GO TO 30
      K = 1
      NIN = NIN + 1
      REWIND 4
   10 CONTINUE
      READ(4,1) XX(K),YY(K)
    1 FORMAT(2E16.8)
      IF(EOF(4).NE.0) GO TO 20
      K = K + 1
      GO TO 10
   20 K = K - 1
   30 KM1 = K - 1
   40 T = RANF(DUM)
      IF(T.LT.YY(1)) GO TO 40
      DO 50 I=1,KM1
      IF(T.GE.YY(I).AND.T.LE.YY(I+1)) GO TO 60
   50 CONTINUE
      I = KM1
   60 YP = XX(I) + (XX(I+1)-XX(I))*(T - YY(I))/
     1 (YY(I+1) - YY(I))
      RETURN
      END
```

Subroutine GEOM.- Subroutine GEOM determines the path of each positron. For each particle which escapes the collimator, the first point of intersection with the magnetic field is calculated. The positron is then assumed to follow a circular path until it exits from the field at the second point of intersection of the path and magnetic field. From this point the particle travels in a straight line until it reaches a distance from the center of the field equal to the distance at which the target was placed. The coordinates of this final point are calculated as well as the total path length and transit time.



APPENDIX

```
      SUBROUTINE GEOM
      DIMENSION X(1000),Y(1000)
      COMMON X,Y,ENER,XMAS,VEL,H,RM,RP,COLL,COLR,YP,THETA,
     1 XP,X1,Y1,X2,Y2,XCM,YCM,XCP,YCP,X3,Y3,PL,TIME,IJK,
     2 EMIN,EMAX,NHIT,YC,RC,EFLD,TMIN,TMAX,NMISS,MODE
      IF(MODE.NE.0) THETA = THETA/57.29577951
      IF(MODE.NE.0) YC = 20.
      YE = COLL*TAN(THETA) + YP
      X(1) = -1.
      IF(ABS(YE).GT.COLR) RETURN
      X(1) = 0.
      D = RM + COLL
      SECT = 1. + TAN(THETA)**2
      RAD = RM*RM*SECT - (YP + D*TAN(THETA))**2
      X1 = (D - YP*TAN(THETA) - SQRT(RAD))/SECT
      Y1 = X1*TAN(THETA) + YP
      XCM = D
      YCM = 0.
      XCP = X1 - RP*SIN(THETA)
      YCP = Y1 + RP*COS(THETA)
      ZK = 2.*(D - XCP)
      ZL = XCP**2 + YCP**2 + RM**2 - RP**2 - D**2
      ZS = ZK**2 + 4.*YCP**2
      ZT = 2.*ZK*ZL - 8.*D*YCP**2
      ZU = ZL**2 + 4.*YCP**2*(D**2 - RM**2)
      X2 = (-ZT + SQRT(ZT**2 - 4.*ZS*ZU))/(2.*ZS)
      Y2 = SQRT(RM**2 - (X2 - D)**2)
      DEL = RP**2 - (X2 - XCP)**2
      IF(DEL.LT.0.00000001) GO TO 10
      SLOPE = (XCP - X2)/SQRT(DEL)
      IF(Y2.LT.YCP) SLOPE = -SLOPE
      B = Y2 - SLOPE*X2
      Y3 = YC
      X3 = (Y3 - B)/SLOPE
      GO TO 20
   10 X3 = X2
      Y3 = YC
   20 CONTINUE
      DIST = (X2 - X1)**2 + (Y2 - Y1)**2
      PHI = ACOS((2.*RP**2 - DIST)/(2.*RP**2))
      PL = PHI*RP
      PL = PL + SQRT(X1**2 + (Y1 - YP)**2)
      PL = PL + SQRT((X3 - X2)**2 + (Y3 - Y2)**2)
      TIME = PL/VEL
      RETURN
      END
```

Subroutine PRT.- Subroutine PRT generates two disk files which may be used as input to the program when operating in a different mode. If the final coordinates of the positron are within the target area, the energy, initial position within the collimator, initial path angle, transit time, and distance from the center of the target are output to TAPE2. For those particles which miss the target, the information is output to TAPE1.




```
      SUBROUTINE PRT
      DIMENSION X(1000),Y(1000)
      COMMON X,Y,ENER,XMAS,VEL,H,RM,RP,COLL,COLR,YP,THETA,
     1 XP,X1,Y1,X2,Y2,XCM,YCM,XCP,YCP,X3,Y3,PL,TIME,IJK,
     2 EMIN,EMAX,NHIT,YC,RC,EFLD,TMIN,TMAX,NMISS,MODE
      DX = X3 - COLL - RM
      IF(ABS(DX).LT.RC) GO TO 10
C   A MISS
      THET = THETA*57.295777951
      WRITE(1) ENER,YP,THET,TIME,DX
      NMISS = NMISS + 1
      GO TO 20
   10 CONTINUE
C   A HIT
      THET = THETA*57.295777951
      NHIT = NHIT + 1
      IF(ENER.LT.EMIN) EMIN = ENER
      IF(ENER.GT.EMAX) EMAX = ENER
      IF(TIME.LT.TMIN) TMIN = TIME
      IF(TIME.GT.TMAX) TMAX = TIME
      WRITE(2) ENER,YP,THET,TIME,DX
   20 RETURN
      END
```

<u>Subroutine PLT</u>.- Subroutine PLT produces graphic output. The plot includes the collimator, the magnetic field, and the trajectory of each positron.



APPENDIX

```
      SUBROUTINE PLT
      DIMENSION X(1000),Y(1000)
      COMMON X,Y,ENER,XMAS,VEL,H,RM,RP,COLL,COLR,YP,THETA,
     1 XP,X1,Y1,X2,Y2,XCM,YCM,XCP,YCP,X3,Y3,PL,TIME,IJK,
     2 EMIN,EMAX,NHIT,YC,RC,EFLD,TMIN,TMAX,NMISS,MODE
      DATA JIK/0/
      JIK = JIK + 1
      IF(JIK.GT.1) GO TO 30
      XMIN = 0.
      YMIN = -10.
      XSF = 3.
      YSF = 3.
      X(1) = 0.
      Y(1) = -COLR
      X(2) = 0.
      Y(2) = COLR
      X(3) = COLL
      Y(3) = COLR
      X(4) = COLL
      Y(4) = -COLR
      X(5) = 0.
      Y(5) = -COLR
      X(6) = XMIN
      Y(6) = YMIN
      X(7) = XSF
      Y(7) = YSF
      CALL LINPLT(X,Y,5,1,0,0,0,0)
      XX = COLL
      DX = 0.05
      IK = 2
      X(1) = XX
      Y(1) = 0.
      XF = COLL + 2.*RM
   10 CONTINUE
      XX = XX + DX
      IF(XX.GT.XF) XX = XF
      YY = SQRT(RM**2 - (XX - XCM)**2)
      X(IK) = XX
      Y(IK) = YY
      IK = IK + 1
      IF(XX.NE.XF) GO TO 10
      XX = XF
   20 CONTINUE
      XX = XX - DX
      IF(XX.LT.COLL) XX = COLL
      YY = -SQRT(RM**2 - (XX - XCM)**2)
      X(IK) = XX
      Y(IK) = YY
      IK = IK + 1
      IF(XX.NE.COLL) GO TO 20
      IK = IK - 1
      X(IK+1) = XMIN
      Y(IK+1) = YMIN
      X(IK+2) = XSF
      Y(IK+2) = YSF
      CALL LINPLT(X,Y,IK,1,0,0,0,0)
```

```
 30 CONTINUE
    X(1) = XP
     Y(1) = YP
    X(2) = X1
    Y(2) = Y1
    IK = 3
    YY = Y1
 40 CONTINUE
    YY = YY + DX
    IF(YY.GT.Y2) YY = Y2
    XX = SQRT(RP**2 - (YY - YCP)**2) + XCP
    X(IK) = XX
    Y(IK) = YY
    IK = IK + 1
    IF(YY.NE.Y2) GO TO 40
    X(IK) = X3
    Y(IK) = Y3
    X(IK+1) = XMIN
    Y(IK+1) = YMIN
    X(IK+2) = XSF
    Y(IK+2) = YSF
    CALL LINPLT(X,Y,IK,1,0,0,0,0)
    RETURN
    END
```

<u>Other subroutines</u>.- The subroutines PSEUDO, NFRAME, CALPLT, and LINPLT are part of the NASA LaRC graphics output library.

Program Usage

Input

All program input is accomplished using FORTRAN list directed reads. The data may appear anywhere in the field and when more than one item is specified, the data items are separated by commas. The first item expected by the program is the variable MODE. Allowed values are 0, 1, and 2. When MODE = 0, the program randomly generates positrons, determines their path through the magnetic field, counts the number of particles which strike the target, and, for those which do strike the target, determines the minimum and maximum energies and transit times. When MODE = 1 or MODE = 2 the program produces graphic output. For MODE = 1, positron trajectories from an earlier MODE = 0 run are plotted. For MODE = 2, the user must specify the initial conditions of the positrons to be plotted.

Additional data are a function of the selected mode of operation. When MODE = 0, the program expects four additional values, separated by commas. The first value is the energy, in kilo-electron-volts, for which focusing is desired. The program uses this to determine the strength of the magnetic field required to focus any positron of this energy on the center of the target. The second value is the distance, in centimeters, from the center of the magnetic field at which the target is to be placed. The third value is the radius of the target, in centimeters. The final value is the number of positrons to be generated. The following is an example of acceptable data for MODE = 0:

```
300.,16.,0.5,5000
```

For MODE = 1, the user must transfer the data on TAPE1 (misses) or TAPE2 (hits) from a MODE = 0 run to TAPE7. The program then expects two additional data values. The first value is the number of paths to plot. The second value is the energy, in kilo-electron-volts, for which focusing is desired. An example follows:

```
4,300.
```

For MODE = 2, the program expects the same two data items as for MODE = 1. Additionally, for each path to be plotted the user must provide three items: the positron energy, in kilo-electron-volts; the initial y-coordinate of the positron within the collimator, in centimeters (-COLR $\leqq$ YP $\leqq$ COLR); and the initial path angle, in degrees ($-2.9^{\circ} \leqq \theta \leqq 2.9^{\circ}$). Sample MODE = 2 input follows:

```
3,300.

250.,0.,-1.

350.,0.,1.

300.,0.,0.
```

Output

Program output also varies with the mode of operation. For MODE = 0, printed output in the following form is generated:

```
      196 HITS          4804 MISSES          5000 TRYS
MINIMUM ENERGY =     .28077858E+03      MAXIMUM ENERGY =      .32205860E+03
MINIMUM TIME =      .12036213E-08     MAXIMUM TIME =      .12665825E-08
```

The energies are in kilo-electron-volts and the times in seconds. For MODE $\neq$ 0, no printed output is produced, but plots are drawn showing the collimator, the magnetic field, and the positron paths as illustrated in figure 12.

APPENDIX

PROGRAMS FOR GENERATING ENERGY AND POSITION DISTRIBUTIONS

Three programs have been written which generate the probability distribution tables required by program POSTRN. Listings of these programs, GENE, GENP, and CUMU, are included in this section of the appendix.

Program GENE tabulates the positron spectrum given by the following expression (ref. 10) at intervals of 2.5 keV:

$$N(W)\ dW = \frac{|P|^2}{\tau_o} F(Z,W)\,(W^2 - 1)^{1/2}(W_o - W)^2\ W\ dW$$

where F(Z,W) for Na^{22} has been calculated using the numerical tables given in reference 11. Values of F(Z,W) for 25 energy values are given in the GENE program listing. The resulting spectrum is illustrated in figure 13.

Program GENP tabulates the probability of a positron being emitted as a function of position within the collimator. It is assumed that the source is shaped like a segment of a sphere with a radius equal to the collimator radius and a height equal to 25% of the collimator diameter. Results are tabulated at 100 equally spaced intervals. The resulting distribution is illustrated in figure 14.

Program CUMU takes as input the output of either program GENE or program GENP. Output from CUMU consists of arrays of normalized cumulative probability distributions versus energy and position. Output from CUMU is supplied on TAPE3 and TAPE4 as input to program POSTRN.

APPENDIX

```
PROGRAM GENE(OUTPUT,TAPE6=OUTPUT,INPUT,TAPE5=INPUT)
DIMENSION E(25),F(25),X(300),Y(300)
COMMON/JTB/NFR,JREQ,IBAUD,JCCS,IJO,TFAC,IJTB(4)
NFR = 0
JREQ = 0
IBAUD = 120
JCCS = 0
TFAC = 0.016
E(1) = 10.1
E(2) = 15.7
E(3) = 22.5
E(4) = 30.4
E(5) = 39.4
E(6) = 49.3
E(7) = 60.3
E(8) = 72.2
E(9) = 84.9
E(10) = 98.4
E(11) = 112.7
E(12) = 127.7
E(13) = 143.4
E(14) = 159.6
E(15) = 176.5
E(16) = 193.8
E(17) = 211.6
E(18) = 248.6
E(19) = 287.2
E(20) = 327.1
E(21) = 368.1
E(22) = 410.2
E(23) = 453.1
E(24) = 496.8
E(25) = 541.1
F(1) = 0.2525
F(2) = 0.3408
F(3) = 0.4122
F(4) = 0.4661
F(5) = 0.5106
F(6) = 0.5481
F(7) = 0.5760
F(8) = 0.6017
F(9) = 0.6222
F(10) = 0.6414
F(11) = 0.6551
F(12) = 0.6684
F(13) = 0.6797
F(14) = 0.6907
F(15) = 0.6988
F(16) = 0.7058
F(17) = 0.7130
F(18) = 0.7231
F(19) = 0.7319
F(20) = 0.7391
F(21) = 0.7449
F(22) = 0.7498
F(23) = 0.7539
F(24) = 0.7574
F(25) = 0.7602
W0 = 2.066
```

```
      ENER = 12.5
      K = 0
      XSUM = 0.
      YSUM = 0.
   10 CONTINUE
      T = ENER/1000.
      W = (T + 0.511)/0.511
      XN = W**2 - 1.
      XN = W*SQRT(XN)*(WO - W)**2
      DO 20 I=1,24
      IF(ENER.GE.E(I).AND.ENER.LE.E(I+1)) GO TO 30
   20 CONTINUE
      I = 24
   30 CONTINUE
      S = F(I) + ((ENER - E(I))/(E(I+1) - E(I)))*(F(I+1) - F(I))
      XN = S*XN
      WRITE(6,1) ENER,XN
      K = K + 1
      X(K) = ENER
      Y(K) = XN*91.538
      XSUM = XSUM + XN
      YSUM = YSUM + XN*ENER
    1 FORMAT(2E16.8)
      ENER = ENER + 2.5
      IF(ENER.LE.540.) GO TO 10
      XMIN = 0.
      XSF = 60.
      X(K+1) = XMIN
      X(K+2) = XSF
      CALL PSEUDO
      Y(K+1) = 0.
      Y(K+2) = 10.
      CALL CALPLT(1.5,1.5,-3)
      CALL AXES(0.,0.,0.,10.,XMIN,XSF,1.,0.,
     1 6HENERGY,0.2,-6)
      CALL AXES(0.,0.,90.,10.,Y(K+1),Y(K+2),1.,0.,
     1 9HFREQUENCY,0.2,9)
      CALL LINPLT(X,Y,K,1,0,0,0,0)
      CALL NFRAME
      CALL CALPLT(0.,0.,999)
      ENAVG = YSUM/XSUM
      WRITE(6,1) ENAVG
      STOP
      END
```

```
      PROGRAM GENP(OUTPUT,TAPE6=OUTPUT,INPUT,TAPE5=INPUT)
      DIMENSION X(300),Y(300)
      COMMON/JTB/NFR,JREQ,IBAUD,JCCS,IJO,TFAC,IJTB(4)
      NFR = 0
      JREQ = 0
      IBAUD = 120
      JCCS = 0
      TFAC = 0.016
      K = 0
      COLR = 0.152*2.54/2.
      A = COLR
      B = A/4.
      BB = (B**2 - A**2)/(2.*B)
      RSQ = A**2 + BB**2
      DIST = -A
      DDIST = A/100.
   10 CONTINUE
      FDIST = SQRT(RSQ - DIST**2) + BB
      WRITE(6,1) DIST,FDIST
      K = K + 1
      X(K) = DIST
      Y(K) = 767.6*FDIST
      DIST = DIST + DDIST
      IF(DIST.LE.A) GO TO 10
    1 FORMAT(2E16.8)
      XMIN = -0.2
      XSF = 0.04
      X(K+1) = XMIN
      X(K+2) = XSF
      CALL PSEUDO
      Y(K+1) = 0.
      Y(K+2) = 10.
      CALL CALPLT(1.5,1.5,-3)
      CALL AXES(0.,0.,0.,10.,XMIN,XSF,1.,0.,
     1 8HPOSITION,0.2,-8)
      CALL AXES(0.,0.,90.,10.,Y(K+1),Y(K+2),1.,0.,
     1 9HFREQUENCY,0.2,9)
      CALL LINPLT(X,Y,K,1,0,0,0,0)
      CALL NFRAME
      CALL CALPLT(0.,0.,999)
      STOP
      END
```

APPENDIX

```
      PROGRAM CUMU(OUTPUT,TAPE6=OUTPUT,TAPE1)
      DIMENSION X(1000),Y(1000)
      K = 1
   10 CONTINUE
      READ(1,1) X(K),Y(K)
    1 FORMAT(2E16.8)
      IF(EOF(1).NE.0) GO TO 20
      K = K + 1
      GO TO 10
   20 K = K - 1
      DO 30 I=2,K
      Y(I) = Y(I) + Y(I-1)
   30 CONTINUE
      DO 40 I=1,K
      Y(I) = Y(I)/Y(K)
   40 CONTINUE
      DO 50 I=1,K
      WRITE(6,1) X(I),Y(I)
   50 CONTINUE
      STOP
      END
```

REFERENCES

1. Holt, W. H.; and Mock, W., Jr.: Positron Lifetime Technique for Nondestructive Evaluation of Materials. NSWC/DL-TR-3573, U.S. Navy, Dec. 1976. (Available from DDC as AD A038 305.)

2. Holt, W. H.; and Mock, W., Jr.: Effect of Moisture Content on Positron Lifetimes in Narmco 5208 Epoxy Resin. Bull. American Phys. Soc., vol. 22, no. 4, Apr. 1977, p. 651.

3. Singh, Jag J.; Holt, William H.; and Buckingham, Richard D.: Positron Annihilation Studies of Moisture in Graphite-Reinforced Composites. NASA paper presented at Conference on Composites and Advanced Materials (Cocoa Beach, Florida), Jan. 1978.

4. Singh, Jag J.; Holt, William H.; Mock, Willis, Jr.; and Buckingham, Richard D.: Moisture Profile Determination in Composites Using Positron Lifetime Techniques. NASA paper presented at 3rd Annual Conference on Composites and Advanced Materials (Merritt Island, Florida), Jan. 1979.

5. Unnam, J.; and Tenney, D. R.: Analytical Prediction of Moisture Absorption/Desorption in Resin Matrix Composites Exposed to Aircraft Environments. Volume A - Structures and Materials, AIAA/ASME 18th Structures, Structural Dynamics & Materials Conference, Mar. 1977, pp. 227-235.

6. Wu, C. S.; and Moszkowski, S. A.: Beta Decay. Interscience Publ., c.1966.

7. Bhalla, C. P.; and Rose, M. E.: Finite Nuclear Size Effects in β Decay. Phys. Rev., second ser., vol. 128, no. 2, Oct. 15, 1962, pp. 774-778.

8. Bhalla, C. P.: Tables of Electron Radial Functions and Tangents of Phase Shifts for Light Nuclei (Z = 1 through 10). NBS Monogr. 81, U.S. Dep. Commer., Aug. 6, 1964.

9. Bhalla, C. P.; and Rose, M. E.: Table of Electronic Radial Functions at the Nuclear Surface and Tangents of Phase Shifts. ORNL-3207, U.S. At. Energy Comm., Jan. 8, 1962.

10. Evans, Robley D.: The Atomic Nucleus. McGraw-Hill Book Co., Inc., c.1955, p. 558.

11. Tables for the Analysis of Beta Spectra. NBS Appl. Math. Ser. 13, U.S. Dep. Commer., June 2, 1952.

TABLE I.- SUMMARY OF CRITICAL PARAMETERS FOR POSITRONS ARRIVING AT TARGET FOR MAGNETIC FIELD SET FOR 200 keV

[Convex source configuration; 1000 positrons emitted at the source]

Parameter	Target diameter	
	1 cm	2.54 cm
Y_C = 9 cm		
$E_{\beta+}$ (min), keV	175.83	154.91
$E_{\beta+}$ (max), keV	230.75	263.40
τ (min), ps	996.10	982.49
τ (max), ps	1085.64	1118.87
No. of hits	68	174
Y_C = 12 cm		
$E_{\beta+}$ (min), keV	187.34	165.20
$E_{\beta+}$ (max), keV	218.70	247.30
τ (min), ps	1143.53	1123.25
τ (max), ps	1215.44	1241.45
No. of hits	44	140
Y_C = 16 cm		
$E_{\beta+}$ (min), keV	187.94	174.86
$E_{\beta+}$ (max), keV	211.70	230.75
τ (min), ps	1350.55	1320.40
τ (max), ps	1411.32	1436.76
No. of hits	38	103
Y_C = 20 cm		
$E_{\beta+}$ (min), keV	191.72	179.44
$E_{\beta+}$ (max), keV	208.06	224.71
τ (min), ps	1549.43	1518.31
τ (max), ps	1587.12	1610.38
No. of hits	28	77

TABLE II.- SUMMARY OF CRITICAL PARAMETERS FOR POSITRONS ARRIVING AT TARGET FOR MAGNETIC FIELD SET FOR 300 keV

[Convex source configuration; 1000 positrons emitted at the source]

Parameter	Target diameter	
	1 cm	2.54 cm
Y_C = 9 cm		
$E_{\beta+}$ (min), keV	272.96	235.58
$E_{\beta+}$ (max), keV	347.05	398.60
τ (min), ps	900.02	894.96
τ (max), ps	952.90	972.04
No. of hits	74	161
Y_C = 12 cm		
$E_{\beta+}$ (min), keV	277.95	250.92
$E_{\beta+}$ (max), keV	328.33	365.64
τ (min), ps	1030.65	1025.05
τ (max), ps	1081.73	1101.24
No. of hits	58	127
Y_C = 16 cm		
$E_{\beta+}$ (min), keV	285.76	262.36
$E_{\beta+}$ (max), keV	317.88	347.05
τ (min), ps	1210.39	1192.08
τ (max), ps	1253.52	1260.10
No. of hits	44	98
Y_C = 20 cm		
$E_{\beta+}$ (min), keV	288.78	272.24
$E_{\beta+}$ (max), keV	315.12	332.66
τ (min), ps	1381.06	1367.96
τ (max), ps	1425.30	1433.93
No. of hits	31	84

TABLE III.- SUMMARY OF CRITICAL PARAMETERS FOR POSITRONS ARRIVING AT TARGET FOR MAGNETIC FIELD SET FOR 400 keV

[Convex source configuration; 1000 positrons emitted at the source]

Parameter	Target diameter	
	1 cm	2.54 cm
Y_C = 9 cm		
E_{β^+} (min), keV	352.47	314.06
E_{β^+} (max), keV	453.85	511.48
τ (min), ps	851.16	846.36
τ (max), ps	898.29	909.75
No. of hits	31	92
Y_C = 12 cm		
E_{β^+} (min), keV	370.30	336.12
E_{β^+} (max), keV	431.36	472.16
τ (min), ps	970.47	970.47
τ (max), ps	1014.94	1023.05
No. of hits	23	64
Y_C = 16 cm		
E_{β^+} (min), keV	377.53	352.47
E_{β^+} (max), keV	414.33	455.31
τ (min), ps	1133.09	1129.55
τ (max), ps	1175.51	1189.39
No. of hits	19	44
Y_C = 20 cm		
E_{β^+} (min), keV	383.53	358.42
E_{β^+} (max), keV	414.33	434.74
τ (min), ps	1293.56	1288.62
τ (max), ps	1336.08	1342.42
No. of hits	17	30



(a) Convex configuration.



(b) Flat configuration.



(c) Concave configuration.

Figure 1.- Profiles of Na^{22} source deposited at bottom of collimator considered in present study.



Figure 2.- Schematic diagram of source collimator.



Figure 3.- Schematic drawing showing source collimator and magnetic field region.



Figure 4.- Profile of Na^{22} source deposited at bottom of collimator selected for detailed study.



(a) 1-cm-diameter target.



(b) 2.54-cm-diameter target.

Figure 5.- Trajectories of positrons striking target from convex source configuration. Magnetic field set for $E_{\beta+}$ = 200 keV.



(a) 1-cm-diameter target.



(b) 2.54-cm-diameter target.

Figure 6.- Trajectories of positrons striking target from flat source configuration. Magnetic field set for $E_{\beta+}$ = 200 keV.



(a) 1-cm-diameter target.



(b) 2.54-cm-diameter target.

Figure 7.- Trajectories of positrons striking target from concave source configuration. Magnetic field set for $E_{\beta+} = 200$ keV.



(a) R = 0.50 cm.   (b) R = 1.27 cm.

Figure 8.- Energy spectra of positrons arriving at target located at different distances from center of magnetic field set for $E_{\beta+}$ = 200 keV.



(a) R = 0.50 cm. (b) R = 1.27 cm.

Figure 9.- Energy spectra of positrons arriving at target located at different distances from center of magnetic field set for $E_{\beta+}$ = 300 keV.



(a) R = 0.50 cm.   (b) R = 1.27 cm.

Figure 10.- Energy spectra of positrons arriving at target located at different distances from center of magnetic field set for $E_{\beta+}$ = 400 keV.



(a) 1-cm-diameter target.



(b) 2.54-cm-diameter target.

Figure 11.- Trajectories of positrons striking target located 20 cm from center of magnetic field set for $E_{\beta+}$ = 300 keV. Convex source configuration.



Figure 12.- Sample program output at 300 keV.



Figure 13.- Computed Na^{22} positron energy spectrum.



Figure 14.- Computed probability of a positron being emitted as a function of position within collimator.

1. Report No. NASA TM-80179	2. Government Accession No.	3. Recipient's Catalog No.
4. Title and Subtitle COMPUTATIONAL METHODS FOR ANALYZING THE TRANSMISSION CHARACTERISTICS OF A BETA PARTICLE MAGNETIC ANALYSIS SYSTEM		5. Report Date December 1979
		6. Performing Organization Code
7. Author(s) Jag J. Singh		8. Performing Organization Report No. L-13345
		10. Work Unit No. 505-33-43-02
9. Performing Organization Name and Address NASA Langley Research Center Hampton, VA 23665		11. Contract or Grant No.
		13. Type of Report and Period Covered Technical Memorandum
12. Sponsoring Agency Name and Address National Aeronautics and Space Administration Washington, DC 20546		14. Sponsoring Agency Code
15. Supplementary Notes Appendix by Gerald H. Mall, Computer Sciences Corp., Hampton, Virginia		

16. Abstract

Computational methods have been developed to study the trajectories of beta particles (positrons) through a magnetic analysis system as a function of the spatial distribution of the radionuclides in the beta source, size and shape of the source collimator, and the strength of the analyzer magnetic field. On the basis of these methods, the particle flux, their energy spectrum, and source-to-target transit times have been calculated for Na^{22} positrons as a function of the analyzer magnetic field and the size and location of the target. These data will be useful in studies requiring parallel beams of positrons of uniform energy -- such as measurement of the moisture distribution in composite materials. Computer programs for obtaining various trajectories are included as an appendix.

17. Key Words (Suggested by Author(s))		18. Distribution Statement	
Radioactive beta sources Positron spectrum Positron lifetime Moisture in composites Magnetic analyzer Energy resolution Transit-time spread Computer programs		Unclassified - Unlimited Subject Category 61	
19. Security Classif. (of this report) Unclassified	20. Security Classif. (of this page) Unclassified	21. No. of Pages 51	22. Price* $5.25

* For sale by the National Technical Information Service, Springfield, Virginia 22161

National Aeronautics and
Space Administration

Washington, D.C.
20546

Official Business
Penalty for Private Use, $300

SPECIAL FOURTH CLASS MAIL
BOOK

Postage and Fees Paid
National Aeronautics and
Space Administration
NASA-451

U.S.MAIL

NASA

POSTMASTER: If Undeliverable (Section 158
Postal Manual) Do Not Return